

# PPN calculation for scalar-tensor gravity in xAct

## Preliminaries

---

### Load package

This loads the xPPN package into the Mathematica session.

```
In[ * ]:= << xAct`xPPN`

-----

Package xAct`xPerm` version 1.2.3, {2015, 8, 23}
CopyRight (C) 2003-2020, Jose M. Martin-Garcia, under the General Public License .
Connecting to external linux executable ...
Connection established .

-----

Package xAct`xTensor` version 1.1.4, {2020, 2, 16}
CopyRight (C) 2002-2020, Jose M. Martin-Garcia, under the General Public License .

-----

These packages come with ABSOLUTELY NO WARRANTY ; for details type
Disclaimer []. This is free software , and you are welcome to redistribute
it under certain conditions . See the General Public License for details .

-----

** DefMetric : Associating fiducial flat derivative PD to metric .
** DefMetric : Associating fiducial flat derivative PD to metric .
... OptionValue : Unknown option FlatMetric for DefCovD .
** DefMetric : Associating fiducial flat derivative PD to metric .
Rules {1} have been declared as UpValues for BkgInvTetradM4 .
Rules {1} have been declared as UpValues for BkgInvTetradS3 .
Rules {1} have been declared as UpValues for BkgTetradM4 .
Rules {1} have been declared as UpValues for BkgInvTetradM4 .
Rules {1} have been declared as UpValues for BkgTetradS3 .
Rules {1} have been declared as UpValues for BkgInvTetradS3 .
... DefMetric : There are already metrics {BkgMetricM4 } in vbundle TM4.
```

Rules {1} have been declared as UpValues for InvTet .

Rules {1} have been declared as DownValues for TREnergyMomentum .

## Nicer printing

Use this to automatically convert "dollar" indices in xAct before output is displayed.

```
In[ * ]:= $PrePrint = ScreenDollarIndices ;
```

## Make rules from equations

These functions help to create xAct rules, either with automatic raising / lowering of indices or without.

```
In[ * ]:= mkrG[eq_Equal] :=
  MakeRule[Evaluate[List @@ eq], MetricOn -> All, ContractMetrics -> True];
mkr0[eq_Equal] := MakeRule[Evaluate[List @@ eq],
  MetricOn -> None, ContractMetrics -> False];
```

## Object definitions

### Scalar field

The scalar field is defined as a tensor with no indices.

```
In[ * ]:= DefTensor[psi[], {MfSpacetime}, PrintAs -> "ψ";
```

### Background value of the scalar field

For the background value of the scalar field we define a constant.

```
In[ * ]:= DefConstantSymbol[psi0, PrintAs -> "Ψ";
```

### Gravitational constant

Another constant parameter we need to define is the gravitational constant.

```
In[ * ]:= DefConstantSymbol[kappa, PrintAs -> "κ";
```

### Parameter function in the action

The kinetic coupling of the scalar field is defined as a scalar function, which depends on the scalar field.

```
In[ * ]:= DefScalarFunction[omega, PrintAs -> "ω";
```

## Field Equations

These tensors will hold the gravitational (metric and scalar) field equations.

```
In[ ]:= DefTensor[MetEq[-T4α, -T4β], {MfSpacetime}, Symmetric[{1, 2}], PrintAs -> "g"];
DefTensor[ScalEq[], {MfSpacetime}, PrintAs -> "g"];
```

## Constant coefficients

For convenience, we define some constants, which will be used in the ansatz for solving the field equations.

```
In[ ]:= aa[i_] := Module[{sym = Symbol["a" <> ToString[i]]},
  If[! ConstantSymbolQ[sym], DefConstantSymbol[sym,
    PrintAs -> StringJoin["!\(a\_", ToString[i], "\)"]]; Return[sym]]
```

## Field equations

### Metric field equation

Here we define the trace-reversed metric field equations. These are more convenient to solve, since the time-time component contains only derivatives of the time-time component of the metric, and so the equations decouple immediately.

```
In[ ]:= psi[] * RicciCD[-T4α, -T4β] - CD[-T4α][CD[-T4β][psi[]]] -
  PD[-T4α][psi[]] * PD[-T4β][psi[]] * omega[psi[]] / psi[] + omega'[psi[]] * InvMet[T4γ, T4δ] *
  PD[-T4γ][psi[]] * PD[-T4δ][psi[]] * Met[-T4α, -T4β] / (4 * omega[psi[]] + 6) - kappa ^ 2 *
  (EnergyMomentum[-T4α, -T4β] - InvMet[T4γ, T4δ] * EnergyMomentum[-T4γ, -T4δ] *
  Met[-T4α, -T4β] * (omega[psi[]] + 1) / (2 * omega[psi[]] + 3))
```

$$\text{Out[ ]:= } -\kappa^2 \left( \Theta_{\alpha\beta} - \frac{\Theta_{\gamma\delta} \text{ig}^{\gamma\delta} g_{\alpha\beta} (1 + \omega[\psi])}{3 + 2 \omega[\psi]} \right) +$$

$$\psi R[\overset{\circ}{\nabla}]_{\alpha\beta} - \overset{\circ}{\nabla}_{\beta} \overset{\circ}{\nabla}_{\alpha} \psi - \frac{\omega[\psi] \partial_{\alpha} \psi \partial_{\beta} \psi}{\psi} + \frac{\text{ig}^{\gamma\delta} g_{\alpha\beta} \partial_{\gamma} \psi \partial_{\delta} \psi \omega'[\psi]}{6 + 4 \omega[\psi]}$$

Save the equations for later use.

```
In[ ]:= meteqdef = MetEq[-T4α, -T4β] == %;
meteqru = mkr0[meteqdef];
```

### Scalar field equation

In the scalar field equation we have already substituted the Ricci scalar, using the trace of the metric field equations, so that only second order derivatives of the scalar field remain.

```
In[ ]:= (2 * omega[psi[]] + 3) * InvMet[T4α, T4β] * CD[-T4α][CD[-T4β][psi[]]] +
  omega'[psi[]] * InvMet[T4α, T4β] * PD[-T4α][psi[]] * PD[-T4β][psi[]] -
  kappa ^ 2 * InvMet[T4α, T4β] * EnergyMomentum[-T4α, -T4β]
```

```
Out[ ]:= -κ2 Θαβ igαβ + igαβ (3 + 2 ω[ψ]) (∇β○ ∇α○ ψ) + igαβ ∂αψ ∂βψ ω'[ψ]
```

Save the equations for later use.

```
In[ ]:= scaleqdef = ScalEq[] == %;
  scaleqru = mkr0[scaleqdef];
```

## Post-Newtonian expansion

### Rules for the scalar field

When the scalar field is expanded in velocity orders, set the zeroth order to the constant background value, and let the first and third orders vanish.

```
In[ ]:= OrderSet[PPN[psi, 0][], psi0];
  OrderSet[PPN[psi, 1][], 0];
  OrderSet[PPN[psi, 3][], 0];
```

### 3 + 1 split

Split the metric equations into their time and space components.

```
In[ ]:= {#, # /. meteqru} &[MetEq[-T4α, -T4β]];
  ChangeCovD[%, CD, PD];
  Expand[%];
  SpaceTimeSplits[#, {-T4α -> -T3a, -T4β -> -T3b}] & /@ %;
  Expand[%];
  Map[ToCanonical, %, {3}];
  Map[SortPDs, %, {3}];
  meteq31list = %;
  meteq31def = Union[Flatten[MapThread[Equal, %, 2]];
  meteq31ru = Flatten[mkr0 /@ %];
```

Split the scalar equation into its time and space components.

```

In[ * ]:= {#, # /. scaleqru} &[ScaleEq[]];
ChangeCovD [% , CD, PD];
Expand[%];
SpaceTimeSplit [#, {}] & /@ %;
Expand[%];
ToCanonical /@ %;
SortPDs /@ %;
scaleq31list = %;
scaleq31def = Equal @@ %;
scaleq31ru = Flatten[mkrg[%]];

```

---

## Velocity orders

Expand each component of the metric equations into the respective velocity orders.

```

In[ * ]:= Outer[VelocityOrder , meteq31list , Range[0, 4]];
Map[NoScalar , %, {4}];
Expand[%];
Map[ContractMetric [#, OverDerivatives -> True,
  AllowUpperDerivatives -> True] &, %, {4}];
Map[ToCanonical , %, {4}];
Map[SortPDs , %, {4}];
meteqvlist = Simplify [%];
meteqvdef = Union[Flatten[MapThread[Equal , %, 3]]];
meteqvru = Flatten[mkrg /@ %];

```

$$\begin{aligned}
\text{Out}[*] = & \left\{ \begin{aligned} \mathcal{E}_{000}^0 & == 0, \quad \mathcal{E}_{000}^1 == 0, \quad \mathcal{E}_{000}^2 == -\frac{4\kappa^2\rho + 2\kappa^2\rho\omega[\Psi] + \Psi(3 + 2\omega[\Psi])\partial_a\partial^a\mathring{g}_{000}}{2(3 + 2\omega[\Psi])}, \quad \mathcal{E}_{000}^3 == 0, \\ \\ \mathcal{E}_{000}^4 & == \frac{1}{4} \left( -4\kappa^2\rho\Pi + \frac{4\kappa^2\rho\Pi}{3 + 2\omega[\Psi]} - \frac{12\kappa^2\rho}{3 + 2\omega[\Psi]} - 4\kappa^2\rho v_a v^a + 4\Psi\partial_0\partial_a\mathring{g}_{00}^3{}^a - 2\Psi\partial_0\partial_0\mathring{g}_{00}^2{}^a - \right. \\ & 4\partial_0\partial_0\mathring{\psi} - 2\Psi\partial_a\partial^a\mathring{g}_{000}^4 - \Psi\partial_a\mathring{g}_{000}^2\partial^a\mathring{g}_{000}^2 - \Psi\partial_a\mathring{g}_{00}^2\partial^b\mathring{g}_{000}^2 - 2\partial_a\mathring{\psi}\partial^a\mathring{g}_{000}^2 + 2\Psi\partial^a\mathring{g}_{000}^2\partial_b\mathring{g}_{00}^2{}^b + \\ & 4\kappa^2\rho\mathring{g}_{000}^2 - \frac{4\kappa^2\rho\mathring{g}_{000}^2}{3 + 2\omega[\Psi]} + 2\Psi\partial^b\partial^a\mathring{g}_{000}^2\mathring{g}_{ab} - 2\partial_a\partial^a\mathring{g}_{000}^2\mathring{\psi} - \frac{2\partial_a\mathring{\psi}\partial^a\mathring{\psi}\omega'[\Psi]}{3 + 2\omega[\Psi]} - \frac{8\kappa^2\rho\mathring{\psi}\omega'[\Psi]}{(3 + 2\omega[\Psi])^2} + \\ & \left. \frac{4\kappa^2\rho\mathring{\psi}\omega'[\Psi]}{3 + 2\omega[\Psi]} + \frac{4\kappa^2\omega[\Psi](\rho\Pi(3 + 2\omega[\Psi]) - 3(3 + 2\omega[\Psi])\rho - \rho(3 + 2\omega[\Psi])\mathring{g}_{000}^2 - 2\rho\mathring{\psi}\omega'[\Psi])}{(3 + 2\omega[\Psi])^2} \right), \\ \\ \mathcal{E}_{0a}^0 & == 0, \quad \mathcal{E}_{0b}^0 == 0, \quad \mathcal{E}_{0a}^1 == 0, \quad \mathcal{E}_{0b}^1 == 0, \quad \mathcal{E}_{0a}^2 == 0, \\ \mathcal{E}_{0b}^2 & == 0, \\ \mathcal{E}_{0a}^3 & == \frac{1}{2} \left( 2\kappa^2\rho v_a - \Psi\partial_0\partial_a\mathring{g}_{00}^2{}^b - 2\partial_0\partial_a\mathring{\psi} + \Psi\partial_0\partial_b\mathring{g}_{00}^2{}^b + \Psi\partial_b\partial_a\mathring{g}_{00}^3{}^b - \Psi\partial_b\partial^b\mathring{g}_{00}^3{}^a \right), \\ \mathcal{E}_{0b}^3 & == \frac{1}{2} \left( 2\kappa^2\rho v_b + \Psi\partial_0\partial_a\mathring{g}_{00}^2{}^a - \Psi\partial_0\partial_b\mathring{g}_{00}^2{}^a - 2\partial_0\partial_b\mathring{\psi} - \Psi\partial_a\partial^a\mathring{g}_{00}^3{}^b + \Psi\partial_a\partial_b\mathring{g}_{00}^3{}^a \right), \\ \mathcal{E}_{0a}^4 & == 0, \quad \mathcal{E}_{0b}^4 == 0, \quad \mathcal{E}_{ab}^0 == 0, \quad \mathcal{E}_{ab}^1 == 0, \\ \mathcal{E}_{ab}^2 & == \frac{1}{6 + 4\omega[\Psi]} \left( -2\kappa^2\delta_{ab}\rho(1 + \omega[\Psi]) + \right. \\ & \left. (3 + 2\omega[\Psi]) \left( \Psi\partial_b\partial_a\mathring{g}_{000}^2 - \Psi\partial_b\partial_a\mathring{g}_{00}^2{}^c - 2\partial_b\partial_a\mathring{\psi} + \Psi\partial_c\partial_a\mathring{g}_{00}^2{}^c + \Psi\partial_c\partial_b\mathring{g}_{00}^2{}^c - \Psi\partial_c\partial^c\mathring{g}_{ab}^2 \right) \right), \\ \mathcal{E}_{ab}^3 & == 0, \quad \mathcal{E}_{ab}^4 == \frac{1}{4} \left( -4\kappa^2\rho v_a v_b - 2\Psi\partial_0\partial_a\mathring{g}_{00}^3{}^b - 2\Psi\partial_0\partial_b\mathring{g}_{00}^3{}^a + 2\Psi\partial_0\partial_0\mathring{g}_{ab}^2 + 2\Psi\partial_b\partial_a\mathring{g}_{000}^4 - \right. \\ & 2\Psi\partial_b\partial_a\mathring{g}_{00}^4{}^c - 4\partial_b\partial_a\mathring{\psi} + \Psi\partial_a\mathring{g}_{000}^2\partial_b\mathring{g}_{000}^2 + \Psi\partial_a\mathring{g}_{00}^2{}^cd\partial_b\mathring{g}_{00}^2{}^cd - \frac{4\omega[\Psi]\partial_a\mathring{\psi}\partial_b\mathring{\psi}}{\Psi} + 2\Psi\partial_c\partial_a\mathring{g}_{00}^4{}^c + \\ & 2\Psi\partial_c\partial_b\mathring{g}_{00}^4{}^c - 2\Psi\partial_c\partial^c\mathring{g}_{ab}^4 + \Psi\partial_a\mathring{g}_{00}^2{}^c\partial_c\mathring{g}_{00}^2{}^d + \Psi\partial_b\mathring{g}_{00}^2{}^c\partial_c\mathring{g}_{00}^2{}^d - \Psi\partial_a\mathring{g}_{00}^2{}^bc\partial^c\mathring{g}_{000}^2 - \\ & \Psi\partial_b\mathring{g}_{00}^2{}^ac\partial^c\mathring{g}_{000}^2 + \Psi\partial_c\mathring{g}_{00}^2{}^ab\partial^c\mathring{g}_{000}^2 - \Psi\partial_c\mathring{g}_{00}^2{}^d\partial^c\mathring{g}_{ab}^2 + 2\partial_a\mathring{g}_{00}^2{}^bc\partial^c\mathring{\psi} + 2\partial_b\mathring{g}_{00}^2{}^ac\partial^c\mathring{\psi} - \\ & 2\partial_c\mathring{g}_{00}^2{}^ab\partial^c\mathring{\psi} - 2\Psi\partial_a\mathring{g}_{00}^2{}^c\partial_d\mathring{g}_{00}^2{}^d - 2\Psi\partial_b\mathring{g}_{00}^2{}^a\partial_d\mathring{g}_{00}^2{}^d + 2\Psi\partial^c\mathring{g}_{00}^2{}^ab\partial_d\mathring{g}_{00}^2{}^d - 2\Psi\partial_c\mathring{g}_{00}^2{}^bcd\partial^d\mathring{g}_{00}^2{}^c + \\ & 2\Psi\partial_d\mathring{g}_{00}^2{}^bc\partial^d\mathring{g}_{00}^2{}^c + 2\Psi\partial_b\partial_a\mathring{g}_{000}^2\mathring{g}_{000}^2 - \frac{4\kappa^2\rho\mathring{g}_{ab}^2}{3 + 2\omega[\Psi]} - \frac{4\kappa^2\rho\omega[\Psi]\mathring{g}_{ab}^2}{3 + 2\omega[\Psi]} + 2\Psi\partial_b\partial_a\mathring{g}_{00}^2{}^cd\mathring{g}_{00}^2{}^cd - \\ & 2\Psi\partial_d\partial_a\mathring{g}_{00}^2{}^bc\mathring{g}_{00}^2{}^cd - 2\Psi\partial_d\partial_b\mathring{g}_{00}^2{}^ac\mathring{g}_{00}^2{}^cd + 2\Psi\partial_d\partial_c\mathring{g}_{00}^2{}^ab\mathring{g}_{00}^2{}^cd + 2\partial_b\partial_a\mathring{g}_{000}^2\mathring{\psi} - 2\partial_b\partial_a\mathring{g}_{00}^2{}^c\mathring{\psi} + \\ & 2\partial_c\partial_a\mathring{g}_{00}^2{}^c\mathring{\psi} + 2\partial_c\partial_b\mathring{g}_{00}^2{}^c\mathring{\psi} - 2\partial_c\partial^c\mathring{g}_{ab}^2\mathring{\psi} - \frac{1}{(3 + 2\omega[\Psi])^2} 2\delta_{ab} \left( 4\kappa^2\omega[\Psi]^2(\rho\Pi - \rho) - \right. \\ & \left. 3\partial_c\mathring{\psi}\partial^c\mathring{\psi}\omega'[\Psi] + 2\omega[\Psi] \left( \kappa^2(5\rho\Pi - 3\rho) - \partial_c\mathring{\psi}\partial^c\mathring{\psi}\omega'[\Psi] \right) + 2\kappa^2\rho \left( 3\Pi + \mathring{\psi}\omega'[\Psi] \right) \right) \left. \right\}
\end{aligned}$$

Expand the scalar equation into the respective velocity orders.

```

In[ * ]:= Outer[VelocityOrder , scaleq31list , Range[0, 4]];
Map[NoScalar , % , {2}];
Expand[%];
Map[ContractMetric [# , OverDerivatives -> True ,
  AllowUpperDerivatives -> True] & , % , {2}];
Map[ToCanonical , % , {2}];
Map[SortPDs , % , {2}];
scaleqvlist = Simplify[%];
scaleqvdef = Flatten[MapThread[Equal , % , 1]]
scaleqvru = Flatten[mkrg /@ %];

```

$$\begin{aligned}
\text{Out[ * ]} = \left\{ \begin{aligned}
& \mathcal{E}^0 == 0, \mathcal{E}^1 == 0, \mathcal{E}^2 == \kappa^2 \rho + (3 + 2 \omega[\Psi]) \partial_a \partial^a \psi^2, \mathcal{E}^3 == 0, \\
& \mathcal{E}^4 == \kappa^2 \rho \Pi - 3 \kappa^2 p - (3 + 2 \omega[\Psi]) \partial_\theta \partial_\theta \psi^2 + (3 + 2 \omega[\Psi]) \partial_a \partial^a \psi^4 - \frac{3}{2} \partial_a \psi^2 \partial^{a2} g_{\theta\theta} - \\
& \omega[\Psi] \partial_a \psi^2 \partial^{a2} g_{\theta\theta} + \frac{3}{2} \partial_a g^2{}_b \partial^a \psi^2 + \omega[\Psi] \partial_a g^2{}_b \partial^a \psi^2 - 3 \partial^a \psi^2 \partial_b g^2{}_a{}^b - 2 \omega[\Psi] \partial^a \psi^2 \partial_b g^2{}_a{}^b - \\
& 3 \partial^b \partial^a \psi^2 g_{ab} - 2 \omega[\Psi] \partial^b \partial^a \psi^2 g_{ab} + \partial_a \psi^2 \partial^a \psi^2 \omega'[\Psi] + 2 \partial_a \partial^a \psi^2 \omega'[\Psi] \}
\end{aligned} \right.
\end{aligned}$$

## Solution

### Check vacuum equations

The zeroth order equations correspond to the vacuum. Check that they are solved for the assumed background.

```

In[ * ]:= eqns0 = {PPN[MetEq, 0][[-LI[0], -LI[0]],
  PPN[MetEq, 0][[-T3a, -T3b], PPN[ScalEq, 0][[]] ] /. meteqvru /. scaleqvru

```

```

Out[ * ]:= {0, 0, 0}

```

### Second order

Extract the second order field equations.

```

In[ * ]:= eqns2 = FullSimplify [
  {PPN[MetEq, 2][[-LI[0], -LI[0]], PPN[MetEq, 2][[-T3a, -T3b], PPN[ScalEq, 2][[]] ] /.
  meteqvru /. scaleqvru]

```

$$\begin{aligned}
\text{Out[ * ]} = \left\{ -\frac{\kappa^2 \rho (2 + \omega[\Psi])}{3 + 2 \omega[\Psi]} - \frac{1}{2} \Psi \partial_a \partial^a \psi^2 g_{\theta\theta}, \frac{1}{6 + 4 \omega[\Psi]} \left( -2 \kappa^2 \delta_{ba} \rho (1 + \omega[\Psi]) + (3 + 2 \omega[\Psi]) \right. \right. \\
\left. \left. (-2 \partial_b \partial_a \psi^2 + \Psi (\partial_b \partial_a \psi^2 g_{\theta\theta} - \partial_b \partial_a g^2{}_c{}^c + \partial_c \partial_a g^2{}_b{}^c + \partial_c \partial_b g^2{}_a{}^c - \partial_c \partial^c g^2{}_{ab})) \right), \kappa^2 \rho + (3 + 2 \omega[\Psi]) \partial_a \partial^a \psi^2 \right\}
\end{aligned}$$

Define an ansatz for the second order metric perturbations.

```
In[ * ]:= ans2def = {PPN[Met, 2][[-LI[0], -LI[0]] == aa[1] * PotentialU [],
  PPN[Met, 2][[-T3a, -T3b] == aa[2] * PotentialU [] * BkgMetricS3 [-T3a, -T3b] +
  aa[3] * PotentialUU [-T3a, -T3b], PPN[psi, 2][[] == aa[4] * PotentialU []]
  ans2ru = Flatten[mkrg /@ ans2def];
```

$$\text{Out[ * ]} = \left\{ g_{00} = a_1 U, g_{ab} = a_2 \delta_{ab} U + a_3 U_{ab}, \psi = a_4 U \right\}$$

Insert the ansatz into the field equations and convert derivatives of the PPN potentials into matter source terms.

```
In[ * ]:= eqns2 /. ans2ru;
PotentialUToChi /@ %;
PotentialUUToChi /@ %;
Expand[%];
ToCanonical /@ %;
ContractMetric[#, OverDerivatives -> True, AllowUpperDerivatives -> True] & /@
  %;
PotentialToSource /@ %;
Expand[%];
ToCanonical /@ %;
SortPDs /@ %;
eqnsa2 = FullSimplify [%]
```

$$\text{Out[ * ]} = \left\{ 2 a_1 \pi \Psi \rho - \frac{\kappa^2 \rho (2 + \omega[\Psi])}{3 + 2 \omega[\Psi]}, \frac{1}{12 + 8 \omega[\Psi]} (-4 \delta_{ab} \rho (\kappa^2 - 6 (a_2 + a_3) \pi \Psi + (\kappa^2 - 4 (a_2 + a_3) \pi \Psi) \omega[\Psi]) + \right. \\ \left. (2 a_4 + (-a_1 + a_2 + a_3) \Psi) (3 + 2 \omega[\Psi]) \partial_c \partial^c \partial_b \partial_a \chi), \rho (\kappa^2 - 12 a_4 \pi - 8 a_4 \pi \omega[\Psi]) \right\}$$

The equations to be solved are extracted as coefficients of the matter terms. The last condition is the gauge condition, which mandates that the spatial part of the metric should be diagonal.

```
In[ * ]:= eqnsc2 = FullSimplify [{Coefficient[eqnsa2[[1]], Density[]],
  Coefficient[eqnsa2[[2]], Density[] * BkgMetricS3 [-T3a, -T3b]],
  Coefficient[eqnsa2[[3]], Density[]], aa[3]]}
```

$$\text{Out[ * ]} = \left\{ 2 a_1 \pi \Psi - \frac{\kappa^2 (2 + \omega[\Psi])}{3 + 2 \omega[\Psi]}, 2 (a_2 + a_3) \pi \Psi - \frac{\kappa^2 (1 + \omega[\Psi])}{3 + 2 \omega[\Psi]}, \kappa^2 - 12 a_4 \pi - 8 a_4 \pi \omega[\Psi], a_3 \right\}$$

Solve for the constant coefficients in the equations.

```
In[ * ]:= sola2 = FullSimplify [First[Solve[#, == 0 & /@ eqnsc2, aa /@ Range[1, 4]]]]
```

$$\text{Out[ * ]} = \left\{ a_1 \rightarrow \frac{2 \kappa^2 + \kappa^2 \omega[\Psi]}{6 \pi \Psi + 4 \pi \Psi \omega[\Psi]}, a_2 \rightarrow \frac{\kappa^2 + \kappa^2 \omega[\Psi]}{6 \pi \Psi + 4 \pi \Psi \omega[\Psi]}, a_3 \rightarrow 0, a_4 \rightarrow \frac{\kappa^2}{12 \pi + 8 \pi \omega[\Psi]} \right\}$$

Check that the solution indeed solves the component equations.

```
In[ * ]:= Simplify[eqnsa2 /. sola2]
```

$$\text{Out[ * ]} = \{0, 0, 0\}$$

Insert the solution for the constant coefficients into the ansatz, to obtain the solution for the metric components.



```
In[ ]:= sol2def = ans2def /. sola2
```

```
sol2ru = Flatten[mkrge /@ sol2def];
```

$$\text{Out[ ]:= } \left\{ \begin{array}{l} \overset{2}{g}_{\theta\theta} == \frac{(2\kappa^2 + \kappa^2 \omega[\Psi]) U}{6\pi\Psi + 4\pi\Psi\omega[\Psi]}, \quad \overset{2}{g}_{ab} == \frac{\delta_{ab} (\kappa^2 + \kappa^2 \omega[\Psi]) U}{6\pi\Psi + 4\pi\Psi\omega[\Psi]}, \quad \overset{2}{\psi} == \frac{\kappa^2 U}{12\pi + 8\pi\omega[\Psi]} \end{array} \right\}$$

Check that the metric components we found indeed solve the second order field equations.

```
In[ ]:= eqns2 /. sol2ru;
```

```
Expand[%];
```

```
PotentialToSource /@ %;
```

```
ToCanonical /@ %;
```

```
SortPDS /@ %;
```

```
Simplify[%]
```

```
Out[ ]:= {0, 0, 0}
```

## Third order

Extract the third order field equations.

```
In[ ]:= eqns3 = FullSimplify[PPN[MetEq, 3][[-LI[0], -T3a] /. meteqvru]
```

$$\text{Out[ ]:= } \kappa^2 \rho v_a - \partial_\theta \partial_a \overset{2}{\psi} + \frac{1}{2} \Psi \left( -\partial_\theta \partial_a \overset{2}{g}^b_b + \partial_\theta \partial_b \overset{2}{g}_a^b + \partial_b \partial_a \overset{3}{g}^b_\theta - \partial_b \partial^b \overset{3}{g}_{\theta a} \right)$$

Define an ansatz for the third order metric perturbations.

```
In[ ]:= ans3def =
```

```
PPN[Met, 3][[-LI[0], -T3a] == aa[5] * PotentialV[-T3a] + aa[6] * PotentialW[-T3a]
```

```
ans3ru = mkrge[ans3def];
```

$$\text{Out[ ]:= } \overset{3}{g}_{\theta a} == a_5 V_a + a_6 W_a$$

Insert the ansatz into the field equations and convert derivatives of the PPN potentials into matter source terms.

```
In[ ]:= eqns3 /. ans3ru /. sol2ru;
```

```
PotentialWToChiV [%];
```

```
Expand[%];
```

```
ContractMetric [%, OverDerivatives -> True, AllowUpperDerivatives -> True];
```

```
PotentialChiToU [%];
```

```
PotentialVToU [%];
```

```
PotentialToSource [%];
```

```
ToCanonical [%];
```

```
SortPDS [%];
```

```
eqnsa3 = FullSimplify [%]
```

$$\text{Out[ ]:= } \frac{(\kappa^2 + 2(a_5 + a_6)\pi\Psi)(4\pi\rho v_a - \partial_\theta \partial_a U)}{4\pi}$$

Solve for the constant coefficients in the equations, keeping a gauge freedom, which is left up to the fourth velocity order.

```
In[ ]:= sola3 = FullSimplify[First[Solve[{eqnsa3 == 0, aa[6] - aa[5] == aa[0]}, {aa[5], aa[6]}]]]
```

$$\text{Out[ ]:= } \left\{ a_5 \rightarrow -\frac{\kappa^2 + 2 a_0 \pi \Psi}{4 \pi \Psi}, a_6 \rightarrow \frac{a_0}{2} - \frac{\kappa^2}{4 \pi \Psi} \right\}$$

Check that the solution indeed solves the component equations.

```
In[ ]:= Simplify[eqnsa3 /. sola3]
```

```
Out[ ]:= 0
```

Insert the solution for the constant coefficients into the ansatz, to obtain the solution for the metric components.

```
In[ ]:= sol3def = ans3def /. sola3
```

```
sol3ru = mkr[sol3def];
```

$$\text{Out[ ]:= } g_{\theta a} = -\frac{(\kappa^2 + 2 a_0 \pi \Psi) V_a}{4 \pi \Psi} + \left( \frac{a_0}{2} - \frac{\kappa^2}{4 \pi \Psi} \right) W_a$$

Check that the metric components we found indeed solve the third order field equations.

```
In[ ]:= eqns3 /. sol2ru /. sol3ru;
```

```
PotentialWToChiV [%];
```

```
Expand[%];
```

```
ContractMetric [%, OverDerivatives -> True, AllowUpperDerivatives -> True];
```

```
PotentialChiToU [%];
```

```
PotentialVToU [%];
```

```
PotentialToSource [%];
```

```
ToCanonical [%];
```

```
SortPDs [%];
```

```
Simplify [%]
```

```
Out[ ]:= 0
```

## Fourth order

Extract the fourth order field equations.

In[ \* ]:= eqns4 = PPN[MetEq, 4][-LI[0], -LI[0]] /. meteqvru

$$\begin{aligned}
 \text{Out[ * ]} = & -\kappa^2 \rho \Pi + \frac{3 \kappa^2 \rho \Pi \omega[\Psi]}{(3 + 2 \omega[\Psi])^2} + \frac{2 \kappa^2 \rho \Pi \omega[\Psi]^2}{(3 + 2 \omega[\Psi])^2} + \frac{\kappa^2 \rho \Pi}{3 + 2 \omega[\Psi]} - \frac{9 \kappa^2 \omega[\Psi] \rho}{(3 + 2 \omega[\Psi])^2} - \frac{6 \kappa^2 \omega[\Psi]^2 \rho}{(3 + 2 \omega[\Psi])^2} - \\
 & \frac{3 \kappa^2 \rho}{3 + 2 \omega[\Psi]} - \kappa^2 \rho v_a v^a + \Psi \partial_0 \partial_a^3 g_0^a - \frac{1}{2} \Psi \partial_0 \partial_0^2 g_a^a - \partial_0 \partial_0^2 \Psi - \frac{1}{2} \Psi \partial_a \partial^a g_{00} - \\
 & \frac{1}{4} \Psi \partial_a^2 g_{00} \partial^a g_{00} - \frac{1}{4} \Psi \partial_a^2 g_b^b \partial^a g_{00} - \frac{1}{2} \partial_a \Psi \partial^a g_{00} + \frac{1}{2} \Psi \partial^a g_{00} \partial_b^2 g_a^b + \\
 & \kappa^2 \rho g_{00}^2 - \frac{3 \kappa^2 \rho \omega[\Psi] g_{00}^2}{(3 + 2 \omega[\Psi])^2} - \frac{2 \kappa^2 \rho \omega[\Psi]^2 g_{00}^2}{(3 + 2 \omega[\Psi])^2} - \frac{\kappa^2 \rho g_{00}^2}{3 + 2 \omega[\Psi]} + \frac{1}{2} \Psi \partial^b \partial^a g_{00}^2 g_{ab} - \\
 & \frac{1}{2} \partial_a \partial^a g_{00}^2 \Psi - \frac{\partial_a \Psi \partial^a \Psi \omega'[\Psi]}{2(3 + 2 \omega[\Psi])} - \frac{2 \kappa^2 \rho \Psi^2 \omega'[\Psi]}{(3 + 2 \omega[\Psi])^2} - \frac{2 \kappa^2 \rho \omega[\Psi] \Psi^2 \omega'[\Psi]}{(3 + 2 \omega[\Psi])^2} + \frac{\kappa^2 \rho \Psi^2 \omega'[\Psi]}{3 + 2 \omega[\Psi]}
 \end{aligned}$$

Define an ansatz for the fourth order metric perturbations.

In[ \* ]:= ans4def =

```

PPN[Met, 4][-LI[0], -LI[0]] == aa[7] * PotentialPhi1 [] + aa[8] * PotentialPhi2 [] +
aa[9] * PotentialPhi3 [] + aa[10] * PotentialPhi4 [] + aa[11] * PotentialU []^2
ans4ru = mkr[ans4def];

```

$$\text{Out[ * ]} = g_{00}^4 == a_7 \Phi_1 + a_8 \Phi_2 + a_9 \Phi_3 + a_{10} \Phi_4 + a_{11} U^2$$

Insert the ansatz into the field equations and convert derivatives of the PPN potentials into matter source terms.

```

In[ * ]:= eqns4 /. ans4ru /. sol2ru /. sol3ru ;
Expand[%];
ContractMetric [%, OverDerivatives -> True, AllowUpperDerivatives -> True];
PotentialVToU [%];
PotentialWToU [%];
PotentialToSource [%];
ToCanonical [%];
SortPDs [%];
Expand [%];
eqns4 = Simplify[ScreenDollarIndices [%]]

```

$$\begin{aligned}
\text{Out[ * ]} = & \frac{1}{32 \pi^2 \Psi (3 + 2 \omega[\Psi])^3} \\
& (-576 \kappa^2 \pi^2 \Psi \rho \Pi + 1728 a_9 \pi^3 \Psi^2 \rho \Pi + 144 \kappa^4 \pi \rho U + 3456 a_{11} \pi^3 \Psi^2 \rho U + 1728 a_8 \pi^3 \Psi^2 \rho U - \\
& 864 \kappa^2 \pi^2 \Psi \rho + 1728 a_{10} \pi^3 \Psi^2 \rho + 864 \pi^2 \Psi (-\kappa^2 + 2 a_7 \pi \Psi) \rho v_a v^a - \\
& 288 \kappa^2 \pi \Psi \partial_0 \partial_0 U + 864 a_0 \pi^2 \Psi^2 \partial_0 \partial_0 U - 48 \kappa^4 \partial_a U \partial^a U - 864 a_{11} \pi^2 \Psi^2 \partial_a U \partial^a U + \\
& 4 \omega[\Psi]^2 (-160 \kappa^2 \pi^2 \Psi \rho \Pi + 576 a_9 \pi^3 \Psi^2 \rho \Pi + 12 \kappa^4 \pi \rho U + 1152 a_{11} \pi^3 \Psi^2 \rho U + \\
& 576 a_8 \pi^3 \Psi^2 \rho U - 384 \kappa^2 \pi^2 \Psi \rho + 576 a_{10} \pi^3 \Psi^2 \rho + 288 \pi^2 \Psi (-\kappa^2 + 2 a_7 \pi \Psi) \rho v_a v^a + \\
& 8 \pi \Psi (-13 \kappa^2 + 36 a_0 \pi \Psi) \partial_0 \partial_0 U - 11 \kappa^4 \partial_a U \partial^a U - 288 a_{11} \pi^2 \Psi^2 \partial_a U \partial^a U) + \\
& 8 \omega[\Psi] (-132 \kappa^2 \pi^2 \Psi \rho \Pi + 432 a_9 \pi^3 \Psi^2 \rho \Pi + 21 \kappa^4 \pi \rho U + 864 a_{11} \pi^3 \Psi^2 \rho U + \\
& 432 a_8 \pi^3 \Psi^2 \rho U - 252 \kappa^2 \pi^2 \Psi \rho + 432 a_{10} \pi^3 \Psi^2 \rho + 216 \pi^2 \Psi (-\kappa^2 + 2 a_7 \pi \Psi) \rho v_a v^a + \\
& 3 \pi \Psi (-25 \kappa^2 + 72 a_0 \pi \Psi) \partial_0 \partial_0 U - 10 \kappa^4 \partial_a U \partial^a U - 216 a_{11} \pi^2 \Psi^2 \partial_a U \partial^a U) + \\
& 8 \omega[\Psi]^3 (-16 \kappa^2 \pi^2 \Psi \rho \Pi + 64 a_9 \pi^3 \Psi^2 \rho \Pi + 128 a_{11} \pi^3 \Psi^2 \rho U + 64 a_8 \pi^3 \Psi^2 \rho U - 48 \kappa^2 \pi^2 \Psi \rho + \\
& 64 a_{10} \pi^3 \Psi^2 \rho + 32 \pi^2 \Psi (-\kappa^2 + 2 a_7 \pi \Psi) \rho v_a v^a + 4 \pi \Psi (-3 \kappa^2 + 8 a_0 \pi \Psi) \partial_0 \partial_0 U - \\
& \kappa^4 \partial_a U \partial^a U - 32 a_{11} \pi^2 \Psi^2 \partial_a U \partial^a U) + 8 \kappa^4 \pi \Psi \rho U \omega'[\Psi] - \kappa^4 \Psi \partial_a U \partial^a U \omega'[\Psi]
\end{aligned}$$

Coefficient of the pressure.

```

In[ * ]:= eq1 = Simplify[Coefficient [eqnsa4, Pressure []]]

```

$$\text{Out[ * ]} = \frac{-3 \kappa^2 + 6 a_{10} \pi \Psi + (-3 \kappa^2 + 4 a_{10} \pi \Psi) \omega[\Psi]}{3 + 2 \omega[\Psi]}$$

Coefficient of the internal energy.

```

In[ * ]:= eq2 = Simplify[Coefficient [eqnsa4, Density [] * InternalEnergy []]]

```

$$\text{Out[ * ]} = \frac{-2 \kappa^2 + 6 a_9 \pi \Psi - (\kappa^2 - 4 a_9 \pi \Psi) \omega[\Psi]}{3 + 2 \omega[\Psi]}$$

Coefficient of the gravitational potential energy.

```

In[ * ]:= eq3 = Simplify[Coefficient [eqnsa4, Density [] * PotentialU []]]

```

$$\begin{aligned}
\text{Out[ * ]} = & \frac{1}{4 \pi \Psi (3 + 2 \omega[\Psi])^3} (18 (\kappa^4 + 12 (2 a_{11} + a_8) \pi^2 \Psi^2) + 3 (7 \kappa^4 + 144 (2 a_{11} + a_8) \pi^2 \Psi^2) \omega[\Psi] + \\
& 6 (\kappa^4 + 48 (2 a_{11} + a_8) \pi^2 \Psi^2) \omega[\Psi]^2 + 64 (2 a_{11} + a_8) \pi^2 \Psi^2 \omega[\Psi]^3 + \kappa^4 \Psi \omega'[\Psi])
\end{aligned}$$

Coefficient of the second time derivative.

In[ ]:= eq4 = Simplify[Coefficient[eqnsa4, ParamD[TimePar, TimePar][PotentialU []]]]

$$\text{Out[ ]:= } - \frac{4 \kappa^2 - 12 a_0 \pi \Psi + 3 \kappa^2 \omega[\Psi] - 8 a_0 \pi \Psi \omega[\Psi]}{12 \pi + 8 \pi \omega[\Psi]}$$

Coefficient of the kinetic energy.

In[ ]:= eq5 = Simplify[Coefficient[eqnsa4, Density[] \* Velocity[-T3a] \* Velocity[T3a]]]

$$\text{Out[ ]:= } -\kappa^2 + 2 a_7 \pi \Psi$$

Coefficient of the potential term.

In[ ]:= eq6 = Simplify[Coefficient[eqnsa4, PD[-T3a][PotentialU []] \* PD[T3a][PotentialU []]]]

$$\text{Out[ ]:= } - \frac{1}{32 \pi^2 \Psi (3 + 2 \omega[\Psi])^3} (48 (\kappa^4 + 18 a_{11} \pi^2 \Psi^2) + 16 (5 \kappa^4 + 108 a_{11} \pi^2 \Psi^2) \omega[\Psi] + 4 (11 \kappa^4 + 288 a_{11} \pi^2 \Psi^2) \omega[\Psi]^2 + 8 (\kappa^4 + 32 a_{11} \pi^2 \Psi^2) \omega[\Psi]^3 + \kappa^4 \Psi \omega'[\Psi])$$

Check that we have fully decomposed the equations.

In[ ]:= Simplify[Pressure[] \* eq1 + Density[] \* InternalEnergy[] \* eq2 + Density[] \* PotentialU[] \* eq3 + ParamD[TimePar, TimePar][PotentialU []] \* eq4 + Density[] \* Velocity[-T3a] \* Velocity[T3a] \* eq5 + PD[-T3a][PotentialU []] \* PD[T3a][PotentialU []] \* eq6 - eqnsa4]

$$\text{Out[ ]:= } 0$$

Solve for the constant coefficients in the equations.

In[ ]:= sola4 = Simplify[First[Solve[#, == 0 & /@ {eq1, eq2, eq3, eq4, eq5, eq6}, aa /@ Prepend[Range[7, 11], 0]]]]]

$$\text{Out[ ]:= } \left\{ \begin{array}{l} a_0 \rightarrow \frac{4 \kappa^2 + 3 \kappa^2 \omega[\Psi]}{12 \pi \Psi + 8 \pi \Psi \omega[\Psi]}, \quad a_7 \rightarrow \frac{\kappa^2}{2 \pi \Psi}, \\ a_8 \rightarrow \frac{\kappa^4 (12 + 38 \omega[\Psi] + 32 \omega[\Psi]^2 + 8 \omega[\Psi]^3 - \Psi \omega'[\Psi])}{16 \pi^2 \Psi^2 (3 + 2 \omega[\Psi])^3}, \quad a_9 \rightarrow \frac{2 \kappa^2 + \kappa^2 \omega[\Psi]}{6 \pi \Psi + 4 \pi \Psi \omega[\Psi]}, \\ a_{10} \rightarrow \frac{3 \kappa^2 + 3 \kappa^2 \omega[\Psi]}{6 \pi \Psi + 4 \pi \Psi \omega[\Psi]}, \quad a_{11} \rightarrow - \frac{\kappa^4 (48 + 80 \omega[\Psi] + 44 \omega[\Psi]^2 + 8 \omega[\Psi]^3 + \Psi \omega'[\Psi])}{32 \pi^2 \Psi^2 (3 + 2 \omega[\Psi])^3} \end{array} \right\}$$

Check that the solution indeed solves the component equations.

In[ ]:= Simplify[eqnsa4 /. sola4]

$$\text{Out[ ]:= } 0$$

Enhance the third order solution, using the gauge fixing condition determined at the fourth order.

In[ ]:= sol3def = ans3def /. Simplify[sola3 /. sola4]

sol3ru = mkr[sol3def];

$$\text{Out[ ]:= } \frac{3}{8} \theta_a = - \frac{\kappa^2 (10 + 7 \omega[\Psi]) V_a}{8 \pi \Psi (3 + 2 \omega[\Psi])} - \frac{\kappa^2 (2 + \omega[\Psi]) W_a}{8 \pi \Psi (3 + 2 \omega[\Psi])}$$

Insert the solution for the constant coefficients into the ansatz, to obtain the solution for the metric components.

```
In[ ]:= sol4def = ans4def /. sola4
```

```
sol4ru = mkrgr[sol4def];
```

$$Out[ ]:= g_{00}^4 == \frac{\kappa^2 \Phi_1}{2 \pi \Psi} + \frac{(2 \kappa^2 + \kappa^2 \omega[\Psi]) \Phi_3}{6 \pi \Psi + 4 \pi \Psi \omega[\Psi]} + \frac{(3 \kappa^2 + 3 \kappa^2 \omega[\Psi]) \Phi_4}{6 \pi \Psi + 4 \pi \Psi \omega[\Psi]} + \frac{\kappa^4 \Phi_2 (12 + 38 \omega[\Psi] + 32 \omega[\Psi]^2 + 8 \omega[\Psi]^3 - \Psi \omega'[\Psi])}{16 \pi^2 \Psi^2 (3 + 2 \omega[\Psi])^3} - \frac{\kappa^4 U^2 (48 + 80 \omega[\Psi] + 44 \omega[\Psi]^2 + 8 \omega[\Psi]^3 + \Psi \omega'[\Psi])}{32 \pi^2 \Psi^2 (3 + 2 \omega[\Psi])^3}$$

Check that the metric components we found indeed solve the fourth order field equations.

```
In[ ]:= eqns4 /. sol2ru /. sol3ru /. sol4ru;
```

```
Expand[%];
```

```
ContractMetric[%, OverDerivatives -> True, AllowUpperDerivatives -> True];
```

```
PotentialVToU [%];
```

```
PotentialWToU [%];
```

```
PotentialToSource [%];
```

```
ToCanonical [%];
```

```
SortPDS [%];
```

```
Expand[%];
```

```
Simplify [%]
```

```
Out[ ]:= 0
```

## PPN metric and parameters

### PPN metric

To read off the PPN parameters, we use the following metric components.

```
In[ ]:= metcomp = {PPN[Met, 2][[-LI[0], -LI[0]], PPN[Met, 2][[-T3a, -T3b],
```

```
PPN[Met, 3][[-LI[0], -T3a], PPN[Met, 4][[-LI[0], -LI[0]]]}
```

$$Out[ ]:= \left\{ g_{00}^2, g_{ab}^2, g_{0a}^3, g_{00}^4 \right\}$$

Insert the solution we obtained into the metric components.

```

In[ ] := metcomp /. sol2ru /. sol3ru /. sol4ru;
ToCanonical [%];
Expand [%];
ppnmet = Simplify [%];
metdef = MapThread[Equal, {metcomp, %}, 1]

```

$$\text{Out[ ]} = \left\{ \begin{aligned}
{}^2g_{00} &== \frac{2\kappa^2 U + \kappa^2 \omega[\Psi] U}{6\pi\Psi + 4\pi\Psi\omega[\Psi]}, & {}^2g_{ab} &== \frac{\kappa^2 \delta_{ab} (1 + \omega[\Psi]) U}{2\pi\Psi (3 + 2\omega[\Psi])}, \\
{}^3g_{0a} &== -\frac{\kappa^2 ((10 + 7\omega[\Psi]) V_a + (2 + \omega[\Psi]) W_a)}{8\pi\Psi (3 + 2\omega[\Psi])}, \\
{}^4g_{00} &== \frac{1}{32\pi^2\Psi^2 (3 + 2\omega[\Psi])^3} \kappa^2 (4\omega[\Psi] (12\pi\Psi (18\Phi_1 + 11\Phi_3 + 21\Phi_4) + \kappa^2 (19\Phi_2 - 20U^2)) + \\
&4\omega[\Psi]^2 (16\pi\Psi (9\Phi_1 + 5\Phi_3 + 12\Phi_4) + \kappa^2 (16\Phi_2 - 11U^2)) + \\
&24(6\pi\Psi (3\Phi_1 + 2\Phi_3 + 3\Phi_4) + \kappa^2 (\Phi_2 - 2U^2)) + \\
&8\omega[\Psi]^3 (8\pi\Psi (2\Phi_1 + \Phi_3 + 3\Phi_4) + \kappa^2 (2\Phi_2 - U^2)) - \kappa^2\Psi (2\Phi_2 + U^2)\omega'[\Psi]) \}
\end{aligned} \right.$$

For later comparison, convert the selected components also to their standard form in terms of PPN parameters and potentials.

```

In[ ] := stamet = Simplify[MetricToStandard /@ metcomp]

```

$$\text{Out[ ]} = \left\{ \begin{aligned}
&2U, 2\gamma \delta_{ba} U, \frac{1}{2} (-(3 + \alpha_1 - \alpha_2 + 4\gamma - 2\xi + \zeta_1) V_a - (1 + \alpha_2 + 2\xi - \zeta_1) W_a), \\
&2\Phi_1 + \alpha_3\Phi_1 + 2\gamma\Phi_1 + \zeta_1(-\mathcal{A} + \Phi_1) + 2\Phi_2 - 4\beta\Phi_2 + 6\gamma\Phi_2 + 2\zeta_2\Phi_2 + \\
&2\Phi_3 + 2\zeta_3\Phi_3 + 6\gamma\Phi_4 + 6\zeta_4\Phi_4 + 2\xi(\mathcal{A} - \Phi_1 + \Phi_2 - 2\Phi_4 - \Phi_w) - 2\beta U^2 \}
\end{aligned} \right.$$

## Newtonian gravitational constant

To solve for the gravitational constant, compare the second order result with the standard normalization.

```

In[ ] := kappaeq = First[ppnmet] == First[stamet]

```

$$\text{Out[ ]} = \frac{2\kappa^2 U + \kappa^2 \omega[\Psi] U}{6\pi\Psi + 4\pi\Psi\omega[\Psi]} == 2U$$

When solving for  $\kappa$ , make sure to catch the positive root.

```

In[ ] := kappadef =
  kappa == First[Sqrt[FullSimplify[k2 /. Solve[kappaeq /. kappa -> Sqrt[k2], k2]]]]
kapparu = mkr[kappadef];

```

$$\text{Out[ ]} = \kappa == 2\sqrt{\pi} \sqrt{\frac{\Psi (3 + 2\omega[\Psi])}{2 + \omega[\Psi]}}$$

## PPN parameters

To generate equations for the PPN parameters, compare the obtained solution with the standard

PPN metric.

`In[ * ]:= pareqns = Simplify[ToCanonical[stamet - ppnmet /. kapparu]]`

$$\text{Out[ * ]} = \left\{ 0, \frac{2 \delta_{ab} (-1 + 2 \gamma + (-1 + \gamma) \omega[\Psi]) U}{2 + \omega[\Psi]}, \right. \\ \left. \frac{1}{2} \left( - \frac{(2(-2 + \alpha_1 - \alpha_2 + 4 \gamma - 2 \xi + \zeta_1) + (-4 + \alpha_1 - \alpha_2 + 4 \gamma - 2 \xi + \zeta_1) \omega[\Psi]) V_a}{2 + \omega[\Psi]} - (\alpha_2 + 2 \xi - \zeta_1) W_a \right), \right. \\ \left. \frac{1}{2(2 + \omega[\Psi])^2 (3 + 2 \omega[\Psi])} \right. \\ \left. (2 \omega[\Psi]^2 (-11 \zeta_1 \mathcal{A} - 18 \Phi_1 + 11 \alpha_3 \Phi_1 + 22 \gamma \Phi_1 + 11 \zeta_1 \Phi_1 - 10 \Phi_2 - 44 \beta \Phi_2 + 66 \gamma \Phi_2 + 22 \zeta_2 \Phi_2 + \right. \\ \left. 22 \zeta_3 \Phi_3 - 54 \Phi_4 + 66 \gamma \Phi_4 + 66 \zeta_4 \Phi_4 + 22 \xi (\mathcal{A} - \Phi_1 + \Phi_2 - 2 \Phi_4 - \Phi_W) + 22 U^2 - 22 \beta U^2) + \right. \\ \left. 4 \omega[\Psi] (-10 \zeta_1 \mathcal{A} - 13 \Phi_1 + 10 \alpha_3 \Phi_1 + 20 \gamma \Phi_1 + 10 \zeta_1 \Phi_1 + \Phi_2 - 40 \beta \Phi_2 + 60 \gamma \Phi_2 + 20 \zeta_2 \Phi_2 + \right. \\ \left. 20 \zeta_3 \Phi_3 - 39 \Phi_4 + 60 \gamma \Phi_4 + 60 \zeta_4 \Phi_4 + 20 \xi (\mathcal{A} - \Phi_1 + \Phi_2 - 2 \Phi_4 - \Phi_W) + 20 U^2 - 20 \beta U^2) + \right. \\ \left. 4 \omega[\Psi]^3 (-\zeta_1 \mathcal{A} - 2 \Phi_1 + \alpha_3 \Phi_1 + 2 \gamma \Phi_1 + \zeta_1 \Phi_1 - 2 \Phi_2 - 4 \beta \Phi_2 + 6 \gamma \Phi_2 + 2 \zeta_2 \Phi_2 + \right. \\ \left. 2 \zeta_3 \Phi_3 - 6 \Phi_4 + 6 \gamma \Phi_4 + 6 \zeta_4 \Phi_4 + 2 \xi (\mathcal{A} - \Phi_1 + \Phi_2 - 2 \Phi_4 - \Phi_W) + 2 U^2 - 2 \beta U^2) + \right. \\ \left. 24 (-\zeta_1 \mathcal{A} - \Phi_1 + \alpha_3 \Phi_1 + 2 \gamma \Phi_1 + \zeta_1 \Phi_1 + \Phi_2 - 4 \beta \Phi_2 + 6 \gamma \Phi_2 + 2 \zeta_2 \Phi_2 + 2 \zeta_3 \Phi_3 - 3 \Phi_4 + \right. \\ \left. 6 \gamma \Phi_4 + 6 \zeta_4 \Phi_4 + 2 \xi (\mathcal{A} - \Phi_1 + \Phi_2 - 2 \Phi_4 - \Phi_W) + 2 U^2 - 2 \beta U^2) + \Psi (2 \Phi_2 + U^2) \omega'[\Psi] \right\}$$

We will consider the coefficients in front of the following potentials.

`In[ * ]:= pots = {PotentialU[] * BkgMetrics3[-T3a, -T3b], PotentialV[-T3a], PotentialW[-T3a], PotentialA[], PotentialU[]^2, PotentialPhiW[], PotentialPhi1[], PotentialPhi2[], PotentialPhi3[], PotentialPhi4[]}`

`Out[ * ]:= {delta_ab U, V_a, W_a, A, U^2, Phi_W, Phi_1, Phi_2, Phi_3, Phi_4}`

Extract the coefficients from the difference between our result and the standard PPN metric. These terms must vanish.

`In[ * ]:= eqns = DeleteCases[Flatten[Simplify[Outer[Coefficient, pareqns, pots]]], 0]`

$$\text{Out[ * ]} = \left\{ \frac{2(-1 + 2 \gamma + (-1 + \gamma) \omega[\Psi])}{2 + \omega[\Psi]}, \right. \\ \left. - \frac{1}{2(2 + \omega[\Psi])} (2(-2 + \alpha_1 - \alpha_2 + 4 \gamma - 2 \xi + \zeta_1) + (-4 + \alpha_1 - \alpha_2 + 4 \gamma - 2 \xi + \zeta_1) \omega[\Psi]), \right. \\ \left. \frac{1}{2} (-\alpha_2 - 2 \xi + \zeta_1), 2 \xi - \zeta_1, \right. \\ \left. (48 - 48 \beta - 80(-1 + \beta) \omega[\Psi] - 44(-1 + \beta) \omega[\Psi]^2 - 8(-1 + \beta) \omega[\Psi]^3 + \Psi \omega'[\Psi]) / (2(2 + \omega[\Psi])^2 (3 + 2 \omega[\Psi])), \right. \\ \left. -2 \xi, \frac{2(-1 + \alpha_3 + 2 \gamma - 2 \xi + \zeta_1) + (-2 + \alpha_3 + 2 \gamma - 2 \xi + \zeta_1) \omega[\Psi]}{2 + \omega[\Psi]}, \right. \\ \left. \frac{1}{(2 + \omega[\Psi])^2 (3 + 2 \omega[\Psi])} (12(1 - 4 \beta + 6 \gamma + 2 \xi + 2 \zeta_2) + (2 - 80 \beta + 120 \gamma + 40 \xi + 40 \zeta_2) \omega[\Psi] + \right. \\ \left. (-10 - 44 \beta + 66 \gamma + 22 \xi + 22 \zeta_2) \omega[\Psi]^2 + 4(-1 - 2 \beta + 3 \gamma + \xi + \zeta_2) \omega[\Psi]^3 + \Psi \omega'[\Psi]), \right. \\ \left. 2 \zeta_3, \frac{2(-3 + 6 \gamma - 4 \xi + 6 \zeta_4 + (-3 + 3 \gamma - 2 \xi + 3 \zeta_4) \omega[\Psi])}{2 + \omega[\Psi]} \right\}$$

List of PPN parameters we are solving for.



```
In[ * ]:= pars = {ParameterBeta , ParameterGamma , ParameterXi ,
  ParameterAlpha1 , ParameterAlpha2 , ParameterAlpha3 ,
  ParameterZeta1 , ParameterZeta2 , ParameterZeta3 , ParameterZeta4 }
```

```
Out[ * ]:= {β, γ, ξ, α1, α2, α3, ζ1, ζ2, ζ3, ζ4}
```

Finally, solve the equations and determine the PPN parameters.

```
In[ * ]:= parsol = FullSimplify[Solve[# == 0 & /@ eqns, pars][[1]]]
```

```
Out[ * ]:= {β → 1 +  $\frac{\Psi \omega'[\Psi]}{4 (2 + \omega[\Psi])^2 (3 + 2 \omega[\Psi])}$ , γ →  $\frac{1 + \omega[\Psi]}{2 + \omega[\Psi]}$ ,
  ξ → 0, α1 → 0, α2 → 0, α3 → 0, ζ1 → 0, ζ2 → 0, ζ3 → 0, ζ4 → 0 }
```